

---

## Picard's Method (Ch. 1, Project B)

---

### PICARD'S METHOD (Ch. 1, Project B)

---

```
> restart;  
> with( plots );  
> with( DEtools );
```

---

Objectives:

- to introduce Picard's method in a manner accessible to students
- to develop a Maple implementation of Picard's method, `picard`
- to use `picard` to motivate discussion of the existence theory for IVPs

New Maple Commands:

`unapply` – converts a Maple expression into a Maple function with selected arguments

---

### Background

First-order initial value problems of the form  $y'(x) = f(x, y(x))$ ,  $y(x_0) = y_0$  can be rewritten as an integral equation

$$y(x) = y_0 + \int_{x_0}^x f(t, y(t)) \, dt.$$

This integral formulation can be used to construct a sequence of approximate solutions to the IVP. The basic idea is that given an initial guess of the approximate solution to the IVP, say  $\phi_0(x)$ , an infinite sequence of functions,  $\{\phi_n(x)\}$ , is constructed according to the rule

$$\phi_{n+1}(x) = y_0 + \int_{x_0}^x f(t, \phi_n(t)) \, dt.$$

That is, the  $n^{\text{th}}$  approximation is inserted into the right-hand side of the integral equation in place of the exact solution  $y(x)$  and used to compute the  $(n+1)^{\text{st}}$  element of the sequence. This process is called Picard's Method.

One of the main applications of Picard's Method is in the proof of the existence and uniqueness results for first-order initial value problems. This proof will not be considered in this report – hey, we have to leave something for next year's students to do. This report focuses on gaining an understanding of Picard's Method through the consideration of a few examples.

---

Example 1: Use Picard's Method with  $\phi_0(x) = 1$  to obtain the next four successive approximations of the solution to  $y'(x) = y(x)$ ,  $y(0) = 1$ .

We start with the given initial approximation to the solution.

```
> phi[0] := 1;
```

$$\phi_0 := 1$$

and compute the next approximation according to the integral formula

```
> phi[1] := 1 + int( phi[0](x), t=0..x );
```

$$\phi_1 := 1 + x$$

The same process is used to compute all subsequent approximations. However, in order to keep the same syntax in the integral, it's necessary to ensure that each  $\phi_n$  is a Maple function. For this the `unapply` function is needed. Thus, the first approximation is

```
> phi[1] := unapply( phi[1], x );
```

$$\phi_1 := x \rightarrow 1 + x$$

and the second approximation to the solution can be obtained by repeating this process

```
> phi[2] := 1 + int( phi[1](t), t=0..x );
```

```
> phi[2] := unapply( phi[2], x );
```

$$\phi_2 := x \rightarrow 1 + x + \frac{1}{2}x^2$$

For each of the last two approximate solutions, the two commands will be combined into a single Maple command

```
> phi[3] := unapply( 1 + int( phi[2](t), t=0..x ), x );
```

$$\phi_3 := x \rightarrow 1 + x + \frac{1}{6}x^3 + \frac{1}{2}x^2$$

```
> phi[4] := unapply( 1 + int( phi[3](t), t=0..x ), x );
```

$$\phi_4 := x \rightarrow 1 + x + \frac{1}{24}x^4 + \frac{1}{2}x^2 + \frac{1}{6}x^3$$


The pattern to the sequence of approximate solutions is obvious. Each iterate adds one new term. The new term for  $\phi_n(x)$  is  $\frac{1}{n!}x^n$ . Thus, for each  $n \geq 0$ ,

$$\phi_n(x) = \sum_{k=0}^n \frac{x^k}{k!}.$$

These functions are the partial sums of the Maclaurin series for  $e^x$ .

From our knowledge of power series, we know that these partial sums converge to the exponential function for all real numbers  $x$ . This convergence can be demonstrated graphically

```
> plot( { exp, phi[i] $ i=0..4 }, 0..3,
>       title='Picard Iterates and Limit for y'=y, y(0)=1' );
```



To conclude this example, we note that the exponential function is a solution to the initial value problem.

---

Example 2: Use Picard's Method with  $\phi_0(x) = 0$  to obtain the next three successive approximations of the solution to the nonlinear problem  $y'(x) = 2x - (y(x))^2$ ,  $y(0) = 0$ .

These approximations could be obtained exactly as in Example 1. However, it will ultimately be more efficient to create a Maple implementation of Picard's Method. The basic ingredients for Picard's Method are the function,  $f$ , on the right-hand side of the ODE, the initial condition, and the initial/previous approximate solution. Here is our version of the method

```
> picard := proc(f, init, phi)
>   local x0, y0;
>   x0 := init[1];
>   y0 := init[2];
>   unapply( y0 + int( f(t, phi(t)), t=x0..x ), x );
> end;
```

Note that the initial condition, `init`, is expected to be the ordered pair  $[x_0, y_0]$  and `f` and `phi` are assumed to be Maple functions. The output from `picard` is the next approximate solution, as a Maple function (so that, *e.g.*, the output is suitable for immediate use in another call to `picard`).

To use `picard` to provide the requested information in this example, let

```
> phi[0] := 0;
```

$$\phi_0 := 0$$

```
> f := (x,y) -> 2*x-y^2;
```

$$f := (x, y) \rightarrow 2x - y^2$$

```
> phi[1] := picard( f, [0,0], phi[0] );
```

$$\phi_1 := (x \rightarrow x)^2$$

(This is simply a fancy way of saying that  $\phi_2(x) = x^2$ .)

```
> phi[2] := picard( f, [0,0], phi[1] );
```

$$\phi_2 := x \rightarrow x^2 - \frac{1}{5}x^5$$

```
> phi[3] := picard( f, [0,0], phi[2] );
```

$$\phi_3 := x \rightarrow x^2 - \frac{1}{5}x^5 + \frac{1}{20}x^8 - \frac{1}{275}x^{11}$$

The problem simply asks for a plot of these approximate solutions on the interval  $0 \leq x \leq 1$ . The picture is a little nicer if a few more iterates are included. The computation of the next two iterates can be done in a simple `for ... do ... od` command. For example,

```
> for n from 0 to 4 do
>   phi[n+1] := picard( f, [0,0], phi[n] );
> od;
> n := 'n':
```

$$\phi_1 := (x \rightarrow x)^2$$

$$\phi_2 := x \rightarrow x^2 - \frac{1}{5}x^5$$

$$\phi_3 := x \rightarrow x^2 - \frac{1}{5}x^5 + \frac{1}{20}x^8 - \frac{1}{275}x^{11}$$

$$\phi_4 := x \rightarrow x^2 + \frac{3}{1540}x^{14} - \frac{87}{374000}x^{17} + \frac{1}{55000}x^{20} - \frac{1}{1739375}x^{23} - \frac{7}{550}x^{11} - \frac{1}{5}x^5 + \frac{1}{20}x^8$$


$$\begin{aligned} \phi_5 := x \rightarrow x^2 &+ \frac{236231}{5298832000000}x^{32} - \frac{6915599}{1734042772000000}x^{35} + \frac{194877}{692159930000000}x^{38} \\ &- \frac{7781}{533431525000000}x^{41} + \frac{1}{2104643750000}x^{44} - \frac{1}{142194993359375}x^{47} \\ &+ \frac{659}{5236000}x^{20} - \frac{1}{5}x^5 + \frac{1}{308}x^{14} - \frac{1769}{2618000}x^{17} + \frac{1}{20}x^8 - \frac{7}{550}x^{11} \\ &- \frac{1363623}{3361446550000}x^{29} - \frac{8751}{413971250}x^{23} + \frac{384503}{123008600000}x^{26} \end{aligned}$$

The convergence of these polynomials is not as obvious as the first example. Even from the graph it's difficult to determine the interval on which this integral exists. Including the direction field in the plot shows precisely how these approximate solutions must behave. Here is the graph that provides the most information. (The plots on  $[0, 1]$  do not show too much difference between the successive approximations. Doubling the interval to  $[0, 2]$  shows a little more of the interesting features of these solutions.

```
> P1 := plot( { phi[i] $ i=0..6 }, 0..2, color=GREEN, thickness=3 );
> P2 := DEplot( diff( y(x), x ) = f(x,y(x)), [x,y], x=0..2, { [0,0] },
>               arrows=THIN );
```

```
> display( { P1, P2 },
>          title='Picard Iterates and Direction Field: y'=2x-y^2, y(0)=0' );
```

Picard Iterates and Direction Field:  $y'=2x-y^2$ ,  $y(0)=0$



Example 3: In previous work it has been demonstrated that  $y'(x) = 3(y(x))^{2/3}$ ,  $y(2) = 0$  does not have a unique solution. Show that Picard's Method, starting with  $\phi_0(x) = 0$  converges to  $y(x) = 0$ , whereas Picard's method beginning with  $\phi_0(x) = x - 2$  converges to the second solution  $y(x) = (x - 2)^3$ .

Each of these can be easily computed using `picard`. But, the first part is so simple, it's really a waste of time and energy to use `picard` or Maple. The claim is that  $\phi_n(x) = 0$  for all  $x$  and for all integers  $n$ . Thus, these approximate solutions converge to the zero solution.

The successive approximations generated when the initial approximation to the solution is

```
> phi[0] := x -> x-2;
```

$$\phi_0 := x \rightarrow x - 2$$

can be computed using a slight variant of `picard`.

The modification is needed because of the rational powers that are involved in the right-hand side of the ODE

```
> f := (x,y) -> 3*y^(2/3);
```

$$f := (x, y) \rightarrow 3y^{2/3}$$

and Maple's inability to choose the branch that leads to the real-valued result. The `combine` function can be used to force some of the simplifications that Maple does not want to make automatically. The modified version of `picard` is

```
> picard2 := proc(f, init, phi)
>   local x0, y0;
>   x0 := init[1];
>   y0 := init[2];
>   unapply( y0 + int( combine( f(t,phi(t)), power ), t=x0..x ), x );
> end;
```

The first two approximates are

```
> phi[1] := picard2( f, [2,0], phi[0] );
```

$$\phi_1 := x \rightarrow \frac{9}{5} (x-2)^{5/3}$$

```
> phi[2] := picard2( f, [2,0], phi[1] );
```

$$\phi_2 := x \rightarrow \frac{27}{95} (x-2)^{19/9} 9^{2/3} 5^{1/3}$$

```
> phi[3] := picard2( f, [2,0], phi[2] );
```

$$\phi_3 := x \rightarrow \int_2^x \frac{3}{95} 27^{2/3} 95^{1/3} \left( (t-2)^{19/9} 9^{2/3} 5^{1/3} \right)^{2/3} dt$$

So, even the modified version of `picard` does not work for all cases. However, it is obvious that the first three approximates all have the form

$$\phi_n(x) = c_n(x-2)^{r_n}$$

for some constants  $c_n$  and  $r_n$ . Explicit formulae for  $c_n$  and  $r_n$  are not apparent from the first three approximate solutions. An alternate approach to the problem of finding these sequences and, hopefully,  $\lim_{n \rightarrow \infty} \phi_n$ , is to obtain a recurrence relation between successive terms in each sequence. To determine the appropriate recurrence relations, suppose the  $n^{\text{th}}$  iterate is

```
> phi[n] := x -> c*(x-2)^r;
```

$$\phi_n := x \rightarrow c(x-2)^r$$

Then, the next iterate will be

```
> phi[n+1] := picard2( f, [2,0], phi[n] );
```

$$\phi_{n+1} := x \rightarrow \lim_{t \rightarrow 2+} 9 \frac{c^{2/3} (t-2) ((t-2)^r)^{2/3}}{3+2r} + 9 \frac{c^{2/3} (x-2) ((x-2)^r)^{2/3}}{3+2r}$$

```
> PHI[n+1] := combine( phi[n+1](x), power );
```

$$PHI_{n+1} := \lim_{t \rightarrow 2+} 9 \frac{c^{2/3} (x-2)^{(1+2/3r)}}{3+2r} + 9 \frac{c^{2/3} (t-2)^{(1+2/3r)}}{3+2r}$$

Maple is hesitant about evaluating the limit because  $r \leq 0$  has not been excluded. However, examination of the first three iterates together with the above result indicates that the exponents will always be positive.

```
> assume( rr>0 );
```

Then, the  $(n+1)^{\text{st}}$  Picard is

```
> eval( subs( r=rr, PHI[n+1] ) );
```

$$9 \frac{c^{2/3} (x-2)^{(1+2/3rr)}}{3+2rr}$$

This provides a recurrence equation for  $r_n$ :

```
> REr := r(n+1) = 1+2/3*r(n);
```

$$REr := r(n+1) = 1 + \frac{2}{3} r(n)$$

The solution to this recurrence equation, with initial condition  $r_0 = 1$ , is found to be

```
> SOLNr := rsolve( { REr, r(0)=1 }, { r(n) } );
```

$$SOLNr := \left\{ r(n) = -2 \left( \frac{2}{3} \right)^n + 3 \right\}$$

Thus, it is now clear that  $\lim_{n \rightarrow \infty} r_n = 3$ . The recurrence equation for the coefficients  $c_n$

```
> REc := c(n+1) = 9/(3+2*r(n))*c(n)^(2/3);
```

$$REc := c(n+1) = 9 \frac{c(n)^{2/3}}{3 + 2r(n)}$$

with initial condition  $c_0 = 1$ , is not so easily solved (by hand or by Maple). It is, nonetheless, still possible to determine the limiting behavior of  $c_n$ . Note that

$$\frac{c_{n+1}}{c_n^{2/3}} = \frac{9}{3 + 2r_n} \rightarrow 1$$

as  $n \rightarrow \infty$ . Assuming that  $c_n$  converges to, say  $c^*$ , then  $c^{*1/3} = 1$  and so  $c^* = 1$ , since these coefficients are real-valued.

This completes the proof that this sequence of Picard iterates converges, with a limit of  $y(t) = \lim_{n \rightarrow \infty} \phi_n(t) = (t - 2)^3$ . It is a simple matter to verify that this function is a solution to the IVP. As a result, this IVP does not have a unique solution. (Note that this conclusion is consistent with the existence and uniqueness theory;  $f_y(x, y)$  is not continuous at  $y = 0$ .)

---

Example 4: Use Picard's method with  $\phi_0(x) = 1$  to compute the next six successive approximations of the solution to the nonlinear problem  $y'(x) = 1 + y(x)^2$ ,  $y(0) = 1/2$ . Create plots of the each Picard iterate superimposed on the direction field. What can you say about the existence of a solution to this IVP?

The data required by `picard` is given as in the previous examples.

```
> f := (x,y) -> 1+y^2;
```

$$f := (x, y) \rightarrow 1 + y^2$$

```
> phi[0] := 1;
```

$$\phi_0 := 1$$

The computation of the Picard iterates is straightforward. Only the first four iterates are displayed, for the sake of saving a few pages of output.

```
> for n from 0 to 5 do
>   phi[n+1] := picard( f, [0,1/2], phi[n] );
> od;
```

```
> phi[1](t);
```

$$\frac{1}{2} + 2t$$

```
> phi[2](t);
```

$$\frac{1}{2} + \frac{5}{4}t + t^2 + \frac{4}{3}t^3$$

```
> phi[3](t);
```

$$\frac{1}{2} + \frac{41}{48}t^3 + \frac{5}{4}t + \frac{5}{8}t^2 + \frac{16}{63}t^7 + \frac{13}{15}t^5 + \frac{23}{24}t^4 + \frac{4}{9}t^6$$

```
> phi[4](t);
```


$$\frac{1}{2} + \frac{24121}{60480}t^9 + \frac{5}{4}t + \frac{5}{8}t^2 + \frac{1808}{36855}t^{13} + \frac{35}{48}t^3 + \frac{10349}{37800}t^{10} + \frac{29}{48}t^4 + \frac{10457}{16128}t^7 + \frac{256}{59535}t^{15}$$

$$+ \frac{223}{320}t^5 + \frac{4157}{5760}t^6 + \frac{3208}{17325}t^{11} + \frac{64}{3969}t^{14} + \frac{5491}{10752}t^8 + \frac{11}{105}t^{12}$$

The convergence of the constant, linear, and quadratic terms are apparent from these results. One additional term seems to be correct after each Picard iterate, but the total number of terms doubles with each iteration.

```
> P1 := DEplot( diff( y(x), x ) = f(x,y(x)), [x,y], x=0..2, {[0,1/2]},
>               y=0..10, arrows=THIN );
> P2 := plot( { phi[i] $ i=0..6 }, 0..2, y=0..10 );
> display( { P1, P2 },
>          title='Direction Field and Picard Iterates: y'=1+y^2, y(0)=1/2' );
```

Direction Field and Picard Iterates:  $y'=1+y^2$ ,  $y(0)=1/2$



While it is not practical to print an animation, the convergence of the Picard iterates to the solution of this IVP is quite clear.

```
> for i from 0 to 6 do
>   p.i := display( { P1, plot( phi[i], 0..2, y=0..10 ) } );
> od;
> i := 'i':
> display( [ seq( p.i, i=0..6 ) ], insequence=true,
>          title='Direction Field and Picard Iterates: y'=1+y^2, y(0)=1/2' );
```

The graphical information also suggests that the solution does not exist for all  $x > 0$ . The solution seems to have a vertical asymptote,  $x = x^*$ , for some  $x^*$  slightly larger than 1. In fact, this problem is easy to solve by hand. The exact solution is

$$y(x) = \tan(x + \arctan(1/2))$$

which has a vertical asymptote at  $x = x^*$  where



```

> xSTAR := Pi/2-arctan(1/2);
                                 $xSTAR := \frac{1}{2}\pi - \arctan\left(\frac{1}{2}\right)$ 
> evalf( xSTAR );
                                1.107148718

```

---

## CONCLUSION

In this project we have developed a Maple implementation of Picard's method and illustrated the use of this procedure for three different initial value problems. These examples have demonstrated the convergence of Picard iterates to solutions of the IVP. The third example is most interesting in that even though the individual successive approximations could not be determined explicitly, the limit of the sequence could still be resolved. The fourth example is an additional problem that is selected to illustrate the convergence of the Picard iterates when the solution exists on only a finite interval.